

## 第 5 章

# 表計算ソフトウェア：少し複雑な処理 (lookup 関数, 絶対参照, IF 関数)

2025 年 6 月 9 日

### 学習目標

- (1) 絶対参照とスピルの考え方を理解する。
- (2) ソートの考え方を理解する。
- (3) IF 関数を使って、条件分けをできるようになる。
- (4) 表引き (LOOKUP 関数) の考え方を理解する。
- (5) さまざまな関数 (ROUNDUP, ABS など) の使い方を理解する。
- (6) 少し複雑な計算式になれる。
- (7) 絶対参照, 相対参照が混じったものやそのスピルでの計算方法を理解する。

本章は、専修大学商学部の高萩栄一郎の著作である。

## 1 表計算ソフトウェアのしくみ

計算式を複製するとき、複製元の計算式中セルの番地は、複製に伴う番地の移動に伴い、複製先の計算式中の番地は自動的に変化します。絶対参照（または絶対番地）とは、複製に伴いセルが移動しても変化させない（固定する）番地の指定方法です。逆に、セルの移動とともに変化させるセルの番地の指定を相対参照（または相対番地）といいます。プログラミングとの対応では、ループの中での定数が絶対参照、ループの中の配列の要素が相対参照に対応します。

表計算ソフトウェアは、複製すると相対参照でアドレスが調整されます。しかし、それでうまくいかないことがあります。そのときは、上記の絶対参照という機能とその範囲の計算式をまとめて設定するスピルという機能があります。

## 2 複製のしくみ

図 1 は、GDP の表を複製で設定した計算式で、次のことがわかります。

**S2:T12(図 1 の青の部分)** 式がかかっているセルの数字が 1 増えると、計算式中の数字も 1 増えています。たとえば、S2 の計算式は「=SUM(B2:R2)」になっており、下に複製した S3 の計算式は「=SUM(B3:R3)」と、すべて数字が 1 増えています。

**B12:R13(図 1 の赤の部分 (一部))** 式が書かれているセルのアルファベットが 1 つ増えると、右に複製された計算式のアルファベットも 1 つ増えています。たとえば、P13 の計算式は「=AVERAGE(P2:P11)」になっており、右に複製した Q13 の計算式は、「=AVERAGE(Q2:Q11)」とアルファベットが 1 つ増えています。

したがって、「複製」という機能は、複製元のセルの番地と複製先のセルの番地との差の分、計算式中の番地を調整していることがわかります。別の言い方をすれば、計算式は設定しているセルからの相対的な位置で記録され、これは相対参照と

	A	P	Q	R	S	T
1	Country	2015	2016	2017	合計	平均
2	Australia	1232.913	1264.524	1379.548	=SUM(B2:R2)	=AVERAGE(B2:R2)
3	Canada	1559.623	1535.768	1653.043	=SUM(B3:R3)	=AVERAGE(B3:R3)
4	China	11226.186	11221.836	12014.61	=SUM(B4:R4)	=AVERAGE(B4:R4)
5	India	2102.392	2273.556	2602.309	=SUM(B5:R5)	=AVERAGE(B5:R5)
6	Japan	4394.977	4950.073	4873.202	=SUM(B6:R6)	=AVERAGE(B6:R6)
7	Korea	1382.764	1414.804	1540.458	=SUM(B7:R7)	=AVERAGE(B7:R7)
8	United Kingdom	2897.06	2669.107	2628.41	=SUM(B8:R8)	=AVERAGE(B8:R8)
9	United States	18219.3	18707.15	19485.4	=SUM(B9:R9)	=AVERAGE(B9:R9)
10	EURO area	11686.259	11960.698	12632.374	=SUM(B10:R10)	=AVERAGE(B10:R10)
11	Asean-5	2047.966	2148.331	2317.146	=SUM(B11:R11)	=AVERAGE(B11:R11)
12	合計	=P2+P3+P4+P5+P6+P7+P8+P9+P10+P11	=Q2+Q3+Q4+Q5+Q6+Q7+Q8+Q9+Q10+Q11	=R2+R3+R4+R5+R6+R7+R8+R9+R10+R11	=SUM(B12:R12)	=AVERAGE(B12:R12)
13	平均	=AVERAGE(P2:P11)	=AVERAGE(Q2:Q11)	=AVERAGE(R2:R11)		

複写すると数字が増えていく

複写するとアルファベットが増えていく

図1 相対参照のしくみ (計算式を表示)

呼ばれています。たとえば、図1のS2の計算式は、「=SUM(B2:R2)」になっています。この計算式は、

=SUM((自分より17つ左のセル)から(自分より1つ左のセル)まで)

というしくみになっています。多くの表は、この複写の番地調整機能(相対参照)でうまくいくのですが、うまくいかないことがあります。その例をやってみましょう。

■参考: 計算式の表示方法 計算式を設定するとその計算結果(数値など)が表示されます。計算結果に替えて、計算式を表示することもできます。

- メニューの「数式」 → ワークシート分析の中の「数式の表示」
- もう一度「数式の表示」をクリックすることにより元に戻る

### 3 絶対参照，相対参照とスピル

#### 3.1 相対参照だけではうまくいかない例

図 2 は，各支店の年度別の売上高を記入したものです．この表は，教科書のページからダウンロードできます (S6.xlsx)．オレンジの背景色の部分は，計算式で入力する部分です

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図 2 売上高の評価

伸び率を計算しましょう．伸び率の計算式は，

$$\frac{2011 \text{ 年の売上高} - 2010 \text{ 年の売上高}}{2010 \text{ 年の売上高}}$$

で，表示形式は「パーセント」です．

次に，2011 年の各支店の売り上げが，合計に占める割合（構成比率）を計算しましょう．

$$\times \text{B5:} = \text{B3/G3}$$

となります。この計算式を B5 から F5 に、複製して見ましょう。おかしい結果になります。原因は表 1 の「単純に複製したときの計算式」（相対参照）になってしまうからです。この場合、C5 から F5 の計算式がおかしいことがわかります。分子は複製先のセルの番地に従って、セルの番地は、相対参照（番地調整機能）によってアルファベットが 1 つずつ増えてよいのですが、分母の番地はいつも「G3」のままでなくてはなりません（青）。

これを解決するために、セル G3 を変化させない方法（絶対参照）と B5 から F5 の計算式をまとめて設定することにより、G3 を変化させない方法があります（スピル機能）状況によりどちらを使うのがよいか異なります。両方使えるようにしておきましょう。

表 1 セルの設定

セル	B5	C5	D5	E5	F5
単純に複製したときの計算式	=B3/G3	=C3/H3	=D3/I3	=E3/J3	=F3/K3
正しい計算式	=B3/G3	=C3/G3	=D3/G3	=E3/G3	=F3/G3

## 3.2 絶対参照

表 1 の正しい計算式のように、複製したときに G3 を変化させないようにするのが絶対参照です。図 3 のように、複製元（B5）のセルを設定するとき、番地調整機能を働かせないようにします。これは、変化させないアルファベットや数字の前に、「\$」マークをつけます。したがって、複製元の計算式は、次式になります。

$$\boxed{\text{B5:}} =\text{B3}/\text{\$G3}$$

この場合、G3 は、常に G3 で、数値の方は変化しないので、

B5: =B3/\$G\$3

としてもかまいません。\$をつけた番地を「絶対参照」、\$を付けない番地を「相対参照」といいます (図 3)。




図 3 絶対参照を使って設定した計算式

■絶対参照の指定の仕方 他の計算式と同様に、キーボードなどから\$マークを入力することが可能ですが、セルをクリックしたあとに、簡単な操作で\$マークを入力する方法があります。

- セルの番地 (G3 など) をクリックなどで入力します。
- キーボードの(F4)キーを押すことにより、\$マークを付けることができます。
- \$ の位置が異なるときは、(F4)キーを何回か押すことにより、指定の計算式にします。

動画: 絶対参照 (音声付き) ・ 動画: 絶対参照 (音声なし)

### 3.3 スピルの利用

Excel 2021 で追加された スピル という機能で、絶対参照を使って指定した範囲の計算式をまとめて指定することができます。まとめて式を設定する範囲を考えれば、絶対参照より直観的に理解できます。

スピルは、デスクトップ版の Excel 2019 などの古いバージョンの Excel では利用できません。その場合、スピルを使うときは、VDI を使用してください。

スピルは、ある範囲のセルの計算をまとめて設定します。

■スピルの図解 図 4 は、伸び率の計算を例です。図 4 上段は、これまでのやり方でやったものです。

B4:	=	(B3-B2)/B2
複写元:		B4
複写先:		C4:G4

通常の利用ではこの方法を利用しましたが、スピルのイメージを把握するため、下段のようにスピルを使って設定します。計算に利用する範囲は 2 つ、計算結果 (出力範囲) が 1 つになります。

範囲 P:	B2:G2	2010 年の値
範囲 Q:	B3:G3	2011 年の値
出力範囲 R:	B4:G4	伸び率
スピルの計算式 (B4 に設定)    =(B3:G3-B2:G2)/B2:G2    B4 に設定すれば、B4:G4 に出力される		

計算は、 $\frac{(2011 \text{ 年の値} - 2010 \text{ 年の値})}{2010 \text{ 年の値}}$  ですので、 $\frac{(Q-P)}{P}$  となります。スピルでは、範囲の各値に対して順番に計算していきます (各範囲の 1 番目の値で計算した結果は、出力範囲の 1 番目に出力)。出力範囲は、入力範囲に応じて自

## 計算式の複写

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)						
6	伸び率評価1						
7	伸び率評価2						

4セルの8%を囲み、赤い矢印で5セルの構成比率(2011)のセルまで伸びる。矢印の途中に「コピー」と書かれている。

5セルの構成比率(2011)のセルに、 $= (B3 - B2) / B2$  と入力されている。

## スピル

	A	B	C	D	E	F	G	
1		A支店	B支店	C支店	D支店	E支店	合計	
2	売上高(2010)	P	1200	1800	2200	1100	1200	7500
3	売上高(2011)	Q	1300	1800	2100	1200	1250	7650
4	伸び率	R	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)							
6	伸び率評価1							
7	伸び率評価2							

B4セルに  $= (B3:G3 - B2:G2) / B2:G2$  と入力されている。この式は、範囲P(売上高2010)と範囲Q(売上高2011)の差を範囲Pで割ることを示している。

範囲P(売上高2010)のセル(2:2~2:7)は緑色で塗りつぶされ、範囲Q(売上高2011)のセル(3:2~3:7)は赤色で塗りつぶされている。

範囲R(伸び率)のセル(4:2~4:7)は青色で塗りつぶされている。

範囲Pのセル(2:2)に  $(Q-P)/P$  と入力されている。

図4 スピルの説明

動的に拡張して表示されます（この拡張するのを、「こぼれ落ちる」ように増やすので、Spillと呼ばれています）。

■スピルの計算式の設定 (B4:G4) 計算式は、次のように設定します。

- (1) B4をクリックし、 $=$ 、 $($ 、 $)$ を入力します。
- (2) マウスもしくはキーボードで B3:G3（範囲Q）を範囲指定し、 $-$ を入力

- (3) B2:G2 (範囲 P) を範囲指定し,  $\text{()}/\text{()}$  を入力
- (4) B2:G2 (範囲 P) を範囲指定し,  $\text{[Enter]}$  を入力
- (5) 計算結果が B4:G4 (範囲 R) に表示されます。

動画: スピルによる設定 (伸び率) (音声付き) ・ 動画: スピルによる設定 (伸び率) (音声なし)

■構成比率の計算 図5は, スピルを使った構成比率の計算です。

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	S 1300	1800	2100	1200	1250	T 7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	U 17%	24%	27%	16%	16%	
6	伸び率評価1	( Sの範囲の各値 ) / Tの値					
7	伸び率評価2	→ Uの範囲の各値				B5:	=B3:F3/G3

図5 スピルによる構成比率の計算

範囲 S:	B3:F3	2011年の値
セル T:	G3	合計の値, 範囲 S の何番目かにかかわらず, いつも G3
出力範囲 U:	B5:F5	構成比率
スピルの計算式 (B5 に設定)	=B3:F3/G3	B5 に設定すれば, B5:F5 に出力される

範囲 S(B3:F3, G3 は含めない) の値を 合計の値のセル (G3) で割って, S の各値の構成比率を求めます。構成比率は,

$B3/G3$ ,  $C3/G3$ , ... のように,  $\text{範囲 } S / \text{セル } T$  になります。分母は, 範囲  $S$  の何番目かにかかわらずいつも  $G3$  です。スピルの場合, 1つのセルを指定した場合, いつもそのセルで計算します (これは, 絶対参照で\$を付けるのに対応します)。

$$B5: =B3:F3/G3$$

- (1) まとめて計算式を指定する範囲の左上のセルを選択します (B5)
- (2) 次の計算式を入力して (B3:F3 は範囲指定を利用した方が便利), Enter

$$B5: =B3:F3/G3$$

- (3) B3:F5 は, B3 から右へ 5 個のセルなので, 計算式を入力した B5 から F5 までの 5 個のセルに こぼれる (spill) ように, 計算結果が表示されます。

[動画: スピルによる設定 \(構成比率\) \(音声付き\)](#) ・ [動画: スピルによる設定 \(構成比率\) \(音声なし\)](#)

## ■注意

- 「#SPILL」「#スピル」というエラーが表示されることがあります。これは, 拡張された出力範囲にすでに値が入力されている場合に表示されます (その範囲を一度クリアするか, 別の場所を指定します)。
- スピルで設定した計算式を変更するときは, 左上のセルの計算式を変更します。

## 4 指数を計算

GDP の計算で, 経済発展が大きい国はどこであるのか分析するとき, 国の大きさに関わらず, GDP がどのように変化しているのかを分析することが重要です。そこで, 物価指数のように基準年を設け, 各国について基準年の値を 100 としたと

き各年の GDP がいくつかになるのかを分析します。

	A	B	C	D	E	F	G	H	I
1	GDP指数 2001年基準	2001	2002	2003	2004	2005	2006	2007	2008
2	Australia	100.00	112.65	143.24	174.14	194.84	207.22	251.38	279.93
3	Canada	100.00	102.93	121.19	138.94	158.80	178.63	198.94	210.36
4	China	100.00	109.93	124.33	146.29	171.78	206.41	265.72	342.56
5	India	100.00	106.04	125.19	146.09	168.89	192.15	250.78	247.83
6	Japan	100.00	95.62	103.30	111.89	110.50	105.27	104.92	117.06
7	Korea	100.00	114.25	127.66	143.49	168.49	189.81	210.61	188.02
8	United Kingdom	100.00	109.20	125.79	147.87	155.44	166.09	189.72	180.46
9	United States	100.00	103.35	108.28	115.42	123.20	130.55	136.57	139.04
10	EURO area	100.00	109.08	134.40	154.05	159.99	169.76	195.69	215.39
11	Asean-5	100.00	113.64	129.07	144.53	160.78	193.70	230.50	270.35
12	全体	100.00	104.77	117.60	130.84	138.86	147.80	164.10	175.99

図6 指数化した表

図6は、その計算表で新しく表（シート名：GDP\_index）を作成しました。計算式は、2001年のデータを100として、その何倍になったかを計算するので、

$$\text{A国}t\text{年の指数(2001年基準)} = \frac{\text{A国}t\text{年のGDP}}{\text{A国}2001\text{年のGDP}} \times 100$$

となります。

■絶対参照による計算 B2の計算式は、ワークシート間の演算を利用して「=(GDP!B2/GDP!B2)\*100」ですが、複写するとき、2000年のGDPはいつもB列なので、Bを変化させないようにBの前に\$をつけ、2の前には付けません。(F4)キー

を使うとき分母の B2 をクリックしたあと、**(F4)**キーを 3 回押すことにより!B2 になります。2000 年から 2007 年までまとめて、計算式を複写します。

<b>B2:</b>	$=(\text{GDP!B2}/\text{GDP!\$B2})*100$
複写元:	B2
複写先:	B2:O13

とします。B 列も同じ計算式で、100 となります。**(F4)**キーは、1 回押す毎に\$マークが付く位置（アルファベットの直前、数字の前）を変更しますので、何回か押して正しい位置になるようにしてください。[動画:Excel GDP の指数](#)

■スピルによる計算 図 7 は、スピルを使った各国地域の指数の計算です。

範囲 V を範囲 W で割り 100 倍する計算を指定しています。V の各値を W のどの値で割るのかはこの指定では自動で調整されます。具体的には、範囲 X の各セルで、分子の範囲 V のセルは、対応セルですが、分母の範囲 W では 1 列のみですので、範囲 X の同じ行の計算式は共通に範囲 W の同じ行の値を使っています。これは絶対参照と同じで、範囲をまとめて設定することにより\$を付ける位置を自動で調整していると考えられることもできます。

- ワークシート GDP\_index を選択（または作成）
- まとめて指定する範囲の左上のセルを選択します (B2)
- B2 に**(≡)**を入力
- GDP のワークシートで、B2:R12 を範囲指定し、**(/)**を入力
- GDP のワークシートで、B2:B12 を範囲指定
- 「\*100」を入力し、**(Enter)**
- 計算式は次のようになります。

ワークシート：GDP

列E～Sは省略

	A	B	C	D	T	U	V	W
1	Country	2001	2002	2003	2019	2020	2021	合計
2	Australia	376.71	424.43	539.56	1,392.33	1,359.37	1,610.56	23,495.45
3	Canada	738.97	760.15	895.60	1,741.58	1,644.04	2,015.98	31,060.89
4	China	1,333.65	1,465.83	1,656.96	14,340.60	14,866.74	16,862.98	161,404.64
5	India	493.93	523.77	618.37	2,870.50	2,660.24	2,946.06	35,451.26
6	Japan	4,374.71	4,182.85	4,519.56	5,135.90	5,045.10	5,103.11	105,453.66
7	Korea	547.74	626.99	702.70	1,651.42	1,638.26	1,823.85	25,780.19
8	United Kingdom	1,639.08	1,782.89	2,052.79	2,833.30	2,709.68	3,108.42	55,110.55
9	United States	10,581.93	10,929.10	11,456.45	21,372.60	20,893.75	22,939.58	339,171.98
10	Euro AREA	6,584.84	7,178.67	8,842.34	13,408.86	12,996.32	14,517.56	250,874.94
11	ASEAN 5	515.88	585.69	664.87	2,734.23	2,602.96	2,821.32	35,847.53
12	合計	27,187.44	28,460.35	31,949.19	67,481.32	66,416.46	73,749.41	1,063,657.08
13	平均	2,718.74	2,846.04	3,194.92	6,748.13	6,641.65	7,374.94	

$(V / W) * 100$

ワークシート：GDP\_index

列E～Sは省略

	A	B	C	D	T	U	V
1	GDP指数 2001年基準	2001	2002	2003	2019	2020	2021
2	Australia	100.00	112.67	143.23	369.60	360.85	427.53
3	Canada	100.00	102.87	121.20	235.68	222.48	272.81
4	China	100.00	109.91	124.24	1075.29	1114.74	1264.42
5	India	100.00	106.04	125.19	581.15	538.58	596.45
6	Japan	100.00	95.61	103.31	117.40	115.32	116.65
7	Korea	100.00	114.47	128.29	301.50	299.09	332.98
8	United Kingdom	100.00	108.77	125.24	172.86	165.32	189.64
9	United States	100.00	103.28	108.26	201.97	197.45	216.78
10	Euro AREA	100.00	109.02	134.28	203.63	197.37	220.47
11	ASEAN 5	100.00	113.53	128.88	530.01	504.56	546.89
12	全体	100.00	104.68	117.51	248.21	244.29	271.26

$(V \text{の範囲の各値}) / (W \text{の範囲の各値}) * 100$   
 → Xの範囲の各値  
 ※Vの同じ行のWは、対応するWの値を利用

GDP\_index!B2:  
 $= (GDP!B2:V12 / GDP!B2:B12) * 100$

図7 スピルを使った各国地域の指数の計算

**B2:** =GDP!B2:R12/GDP!B2:B12\*100

- 計算式で「GDP!B2:R12」とあるので、B列からR列まで17列、2行目から12行目まで11行、式を入力するセルB2から右下へ17列12行まとめて計算します。
- 分母の「GDP!B2:B12」は行のみの変化ですので、行のみの範囲を指定します。
- B2:B12が、2001年基準（B列）を表します。

動画: [スピル・指数の計算（音声付き）](#) ・ [動画: スピル・指数の計算（音声なし）](#)

## 4.1 練習問題 S1-6

スピルまたは絶対参照を利用して作成しましょう。

- (1) ワークシート「GDP\_index2008」を作成し、2008年基準（リーマンショック）のGDPの指数の表を作成しなさい。
- (2) ワークシート「population\_index2001」を作成し、2001年基準の人口の指数の表を作成しなさい。
- (3) ワークシート「GDPperCapita\_index2010」を作成し、2010年基準の一人あたりGDPの指数の表を作成しなさい。

## 5 練習問題

### 5.1 練習問題 S7-1

図8は、関東の各都県の小学生、中学生、高校生、大学生の人数を千人単位で記入したものです。（出典：学校基本調査，1999年5月1日現在）この表は、[教科書のページ](#)からダウンロードできます（S7.xlsx）。

- (1) B10からE10に、小学生、中学生、高校生、大学生の人数の合計を計算しなさい。
- (2) F3に、茨城県の小学生人数の関東の都県の合計に占める割合（パーセント表示）を計算します

	A	B	C	D	E	F	G	H	I
1		人数(単位:1000人)				各都県が占める割合			
2	都県名	小学生	中学生	高校生	大学生	小学生	中学生	高校生	大学生
3	茨城	189	110	106	37	8.55%	8.94%	8.58%	3.20%
4	栃木	128	74	77	21	5.79%	6.01%	6.23%	1.82%
5	群馬	124	69	67	19	5.61%	5.61%	5.42%	1.65%
6	埼玉	408	217	206	114	18.45%	17.63%	16.67%	9.87%
7	千葉	339	190	183	114	15.33%	15.43%	14.81%	9.87%
8	東京	562	329	365	660	25.42%	26.73%	29.53%	57.14%
9	神奈川	461	242	232	190	20.85%	19.66%	18.77%	16.45%
10	合計	2211	1231	1236	1155	100.00%	100.00%	100.00%	100.00%

図8 関東の県の児童生徒学生の比率

絶対参照またはスピルを使って指定しなさい。

絶対参照を使う場合 F3 の計算式を F3 から F9 に複写するために、F3 の計算式の適当な位置に\$マークをつけます。

ただし、\$マークは変化させないものの前につけます。

アルファベットを変化させたくない場合は、アルファベットの、数字を変化させたくない場合は数字の前に\$をつけます。

F3 の計算式を F3 から F9 に複写します。

スピルを使う場合 各都道府県の小中高大学生の合計を sum 関数を使い、計算式の中に含めます。

スピルで、sum(範囲)を使うと、sum(範囲)は1つの値を出力する関数ですので、1つの値として扱われます。

(3) 中学生、高校生、大学生の列も同様に計算式を設定します。

	A	B	C	D	E
1	合否判定				
2	合格基準点	70			
3					
4					
5	番号	名前	得点	平均との差	合否
6	1	田中	100	34.8	1
7	2	大西	70	4.8	1
8	3	大久保	30	-35.2	0
9	4	前田	80	14.8	1
10	5	友田	65	-0.2	0
11	6	保田	77	11.8	1
12	7	大川	78	12.8	1
13	8	木村	32	-33.2	0
14	9	村田	54	-11.2	0
15	10	川上	66	0.8	0
16		合計	652		5
17		平均	65.2		

図9 合否判定

## 5.2 練習問題 S8-1

図9のように入試の合否判定をします。この表は、[教科書のページ](#)からダウンロードできます(S8.xlsx)。ただし、E列の合否判定は、IF関数の節で学習します。

- (1) 得点の合計，平均を計算しましょう。
- (2) 平均との差を計算しましょう。平均値はいつも C17 であることを注意しましょう。  
絶対参照またはスピルを使います。

### 5.3 練習問題 S1-7

各国・地域の GDP が各年，10 ヶ国・地域の合計に対しての割合（構成割合）を求めましょう。

- (1) 新しいシート「GDP\_share」を作成し，指数と同様に，表頭，表側を複製します。
- (2) A 国・地域の t 年の構成割合は，次式で計算します。

$$\text{A 国・地域 t 年の構成割合} = \frac{\text{A 国・地域 t 年の GDP}}{\text{t 年の 10 ヶ国・地域の GDP の合計}}$$

B2～V11 に，A 国・地域 t 年の構成割合を計算します。絶対参照またはスピルを使います。

絶対参照 B2 に計算式を設定します。B2 の場合，分子は 2001 年の Australia の GDP(GDP!B2) になります。分母は，2001 年の GDP の合計 (GDP!B12) になりますが，複製するとき，合計はいつも 12 行目であることに注意しましょう。

複製元: B2      複製先: B2:R11

スピル 分母は，指数の計算では 2001 年の値の列でしたが，構成割合では合計の行になることに注意しましょう。

- (3) 表示形式をパーセンテージにします。
- (4) 図 10 は，計算例です。

	A	B	C	D	E	F	G	H
1	Country	2001	2002	2003	2004	2005	2006	2007
2	Australia	1.39%	1.49%	1.69%	1.85%	1.95%	1.95%	2.13%
3	Canada	2.72%	2.67%	2.80%	2.89%	3.11%	3.29%	3.29%
4	China	4.91%	5.15%	5.19%	5.48%	6.07%	6.86%	7.97%
5	India	1.82%	1.84%	1.94%	2.03%	2.21%	2.36%	2.78%
6	Japan	16.09%	14.70%	14.15%	13.76%	12.81%	11.46%	10.27%
7	Korea	2.01%	2.20%	2.20%	2.23%	2.48%	2.62%	2.63%
8	United Kingdom	6.03%	6.26%	6.43%	6.79%	6.72%	6.75%	6.94%
9	United States	38.92%	38.40%	35.86%	34.36%	34.57%	34.41%	32.46%
10	Euro AREA	24.22%	25.22%	27.68%	28.51%	27.87%	27.81%	28.85%
11	ASEAN 5	1.90%	2.06%	2.08%	2.10%	2.20%	2.49%	2.67%

図 10 GDP の構成割合 (一部)

## 6 ソート (並べ替え) 関数

ソート (ソーティング, 並べ替え) は, ある基準で行 (または列) の順番を入れ換えることです。成績処理をするとき, 得点の大きい順に並べ替えるときなどに使います。GDP の例ではある年の GDP の大きさなどの順番で並べ替えてみます。例題の表は, [教科書のページ](#)からダウンロードできます (sort\_ex.xlsx)。

2023 年度から本テキストでは Sort 関数を利用します。従来のテキストでは, 関数を使わずに, 直接, 表を書き換える方法でした。直接書き換えると, 分析の途中で元の表が書き換わり操作が複雑になったり, 操作ミスにより元にデータが壊れたりしますので, ソート関数を使い, 別の表に出力するようにしました。古いバージョンの Office では, Sort 関数は利用できません。その場合は VDI を利用してください。

## 6.1 ソート関数の利用

Excel ではソートをする関数は、SORT 関数と SORTBY 関数の 2 つあります。SORT 関数は 1 基準のみで、SORTBY 関数は複数基準を扱えます。本テキストでは、複数基準を扱う SORTBY 関数を学修します。

ソート (並べ替え) は、ある基準に従って行を並べ替えます。並べ替えの方法は、大きい順は降順、小さい順 (または ABC 順、あいうえお順) は昇順と呼びます。並べ替えの基準の列ををソートキーもしくは単にキーといいます。

この節の例題は、[教科書のページ](#)からダウンロードできます (sort.ex.xlsx)。図 11 の元データ (出席簿のデータ) で、クラス、出席番号はばらばらの順番です。まず、真ん中の 1 基準のように、クラスの順番 (昇順) で並べ替えてみます。この場合、クラスの順番は整いますが、出席番号はばらばらです。そこで、出席番号の大きい順 (降順) に並べます (通常は、小さい順ですが、ここでは練習のため大きい順にします)。このように、基準となる列が複数指定するとき、基本はクラス順で、クラスが同じ時とは出席番号順に並べ替えるとき、クラス順を 1 番目のソートキーにし、出席番号を 2 番目のソートキーにします。

SORTBY 関数は、元のデータの範囲を指定し、別の範囲にソートした結果を出力します。例えば、図 11 のワークシート **元データ** の範囲 A2:A9 を別の範囲、図 11 のワークシート **1 基準** の A2:C19 に、ソート結果を表示します。関数は、左上のセル A2 に指定すれば、C19 まで自動で拡張 (スピル) して表示されます。**1 基準** の A2:C19 のように、スピルの範囲はスピル範囲と呼ばれています。

SORTBY 関数は次のように指定します。

= SORTBY (もとの範囲, 基準 1 の列の範囲, 基準 1 の順序, 基準 2 の列の範囲, 基準 2 の順序, ...)

**元の範囲** 並べ替える元の範囲 (図 11 の例では、**元データ** の A2:C19)。

**基準 1 の列の範囲** 並べ替える基準の列の範囲。行数は並べ替える元の範囲と同じ行数 (図 11 の **1 基準** の例では、

The figure shows three stages of sorting a table. The table has columns A (クラス), B (出席番号), and C (氏名). The data rows are 2 through 9.

Row	クラス (A)	出席番号 (B)	氏名 (C)
2	B	1	生田花子
3	C	1	秋葉原太郎
4	A	3	神保町四郎
5	A	1	あざみ野菊代
6	B	2	向ヶ丘次郎
7	C	2	水道橋桜美
8	A	2	飯田橋梅香
9	B	3	登戸三郎

1. 元のデータ (Original Data): The table is in its original order.

2. 1基準 (クラス昇順) (1st Criterion: Class Ascending): The rows are sorted by Class (A, B, C). Row 2 (B) is first, followed by rows 5, 6, 7, 8, 9, and 4.

3. 2基準(クラス昇順,出席番号降順) (2nd Criterion: Class Ascending, Attendance Number Descending): Within each class, rows are sorted by Attendance Number in descending order. For Class A, row 5 (1) is first, then row 4 (3). For Class B, row 6 (2) is first, then row 9 (3). For Class C, row 7 (2) is first, then row 3 (1).

元のデータ

1基準 (クラス昇順)

2基準(クラス昇順,出席番号降順)

図 11 ソート関数の図解

元データ の A2:C9).

基準 1 の順序 昇順の場合 1, 降順の場合-1 を指定

図 11 のワークシート 1 基準 の A2 の計算式を設定します。

- (1) ワークシート 1 基準 の A2 をクリックします。
- (2) = 「=SORTBY(」と入力
- (3) = 元データ の A2:C19 を範囲指定し, 「,」を入力

- (4) = 元データ の A2:A19 を範囲指定し、「,」を入力  
 (5) = アルファベット順は昇順なので、「1」を入力  
 (6) = 「)」を入力し、Enterを入力

=SORTBY(元データ!A2:C9, 元データ!A2:A9,1)

となります。

動画:Sortby 関数, 1 基準 (音声付き) ・ 動画:Sortby 関数, 1 基準 (音声なし)

ワークシート 2 基準 の A2 の計算式は、次のようになります。基準 1 と同じ方法で設定してみましょう。手順の (3),(4) をくり返すのと、基準 2 の順序が降順であることに注意してください。

=SORTBY(元データ!A2:C9, 元データ!A2:A9,1, 元データ!B2:B9,-1)

ソート関数の特徴

- ソート関数は、元の表を書き直すと自動的にソート関数で出力されるスピル範囲は変化します。例えば、元データ のクラス記号や出席番号を変更してみましょう。1 基準 や 2 基準 の表が変化します。
- スピル範囲の個別のセルは変更できません。  
 計算式を変更するには、左上のセル (A2) の計算式を変更します。  
 スピル範囲を消すには、スピル範囲全体 (A2:C9) の中をクリックすると青色の枠で選択されるので、そこで Delete キーで削除します。

動画:Sortby 関数, 2 基準 (音声付き) ・ 動画:Sortby 関数, 2 基準 (音声なし)

## 6.2 練習問題 S1-8

ワークシート「GDP\_Sort」を作成し、2001年を基準として大きい順にソートした結果を表示する表を作成しなさい。  
※ソートする範囲は、列はデータの一番右まで含め、行は見出し行や合計や平均の行は含めません（含めると一緒にソートします）。

## 7 RANK 関数を利用して順位を求める

S1.xlsx のワークシート「GDP」の AC 列のように、2001 年の GDP での順位を求めましょう（図 12）。順位は、RANK 関数という関数とスピルという Excel の新しい機能で求めることができます。

※ 表の画面をを書き直す必要あり

### 7.1 RANK 関数

図 13 は、RANK 関数の図解です。RANK 関数には、3 つの引数があります。

**数値** 順位を求める値です。

**範囲** この範囲で指定した範囲の順位を求めます。

**順序** 大きい順（降順）のとき 0、それ以外（小さい順（昇順））のとき 0 以外の値を指定します。

	A	T	U	V	W	X	Y	Z	AA	AB	AC	AD	AE
1	Country	2019	2020	2021	合計	平均	中央値	GDP 2000→2021	平均倍数 幾何平均	平均増減率 (1年あたり増減率)	2001順位	2021順位	順位 平均増減率
2	Australia	1,392.33	1,359.37	1,610.56	23,495.45	1,118.83	1,251.86	4.2753	1.0753	7.53%	10	10	
3	Canada	1,741.58	1,644.04	2,015.98	31,060.89	1,479.09	1,556.51	2.7281	1.0515	5.15%	6	8	
4	China	14,340.60	14,866.74	16,862.98	161,404.64	7,685.94	7,492.21	12.6442	1.1353	13.53%	5	2	
5	India	2,870.50	2,660.24	2,946.06	35,451.26	1,688.16	1,823.05	5.9645	1.0934	9.34%	9	6	
6	Japan	5,135.90	5,045.10	5,103.11	105,453.66	5,021.60	5,003.68	1.1665	1.0077	0.77%	3	4	
7	Korea	1,651.42	1,638.26	1,823.85	25,780.19	1,227.63	1,253.42	3.3298	1.0620	6.20%	7	9	
8	United Kingdom	2,833.30	2,709.68	3,108.42	55,116.55	2,624.60	2,704.50	1.8964	1.0325	3.25%	4	5	
9	United States	21,372.60	20,893.75	22,939.58	339,171.98	16,151.05	15,599.73	2.1678	1.0394	3.94%	1	1	
10	Euro AREA	13,408.86	12,996.32	14,517.56	250,874.94	11,946.43	12,666.69	2.2047	1.0403	4.03%	2	3	
11	ASEAN 5	2,734.23	2,602.96	2,821.32	35,847.53	1,707.03	1,971.23	5.4689	1.0887	8.87%	8	7	
12	合計	67,481.32	66,416.46	73,749.41	1,063,657.08		53,972.31	2.7126	1.0512	5.12%			
13	平均	6,748.13	6,641.65	7,374.94									

図 12 RANK 関数（ワークシート GDP）




図 13 RANK 関数の図解

## 7.2 スピルを使って設定

計算式は、次のようになります。

AC2: =RANK(B2:B11,B2:B12,0)

- RANK 関数の最初の引数は、1つの値なので、最初の引数の B2:B11 は、下に 10 個の分 (Y2:Y11) に計算することを指定します。
- RANK 関数の 2 つ目の引数は、順位を求める範囲なので B2:B11 とします。(特に、「\$」を付ける必要はありません)

図 13 は RANK 関数の図解で RANK 関数は 3 つの引数を使います。AC2 にオーストラリアの 2001 年の順位を求めます。次のように設定します。

- (1) AC2 をクリック
- (2) 「=RANK(」と入力
- (3) B2 から B11 を範囲指定し「,」を入力
- (4) B2 から B11 を範囲指定
- (6) 「)」を入力し、Enterを入力
- (7) AC2 から AC11 に 2001 年の順位が表示されたと思います。

[動画:RANK 関数スピル \(音声付き\)](#) ・ [動画:RANK 関数スピル \(音声なし\)](#)

## 7.3 絶対参照を使って設定

次のように設定します。

- (1) AC2 をクリック
- (2) 「=RANK(」 と入力
- (3) B2 をクリックし, 「,」 を入力
- (4) B2 から B11 をマウスで範囲指定
- (5) キーボードから(F4)キーを 1 回入力. 「B2:B11」 が 「\$B\$2:\$B\$11」 となります.
- (6) 「)」 を入力し, (Enter) を入力

複写

複写元: AC2

複写先: AC3:AC11

[動画:Excel 順位を求める, 関数の挿入を利用](#)

## 8 練習問題

### 8.1 練習問題 S1-9

- (1) AD 列に 2021 の順位を求めましょう.

(2) ワークシート GDPperCapita の AC 列に 2001 年の、AD 列に 2021 年の一人あたりの GDP の順位を求めましょう。

## 8.2 練習問題 S4-2

練習問題 S4-1 で作成したモーグルの表を「トータルスコア」の高い順に順位を付けなさい。

# 9 オートフィルター

オートフィルタは、ある特定の条件を満たす行だけ抽出する機能です。

## 9.1 オートフィルターの設定・解除



オートフィルターを実行すると、大きい順や小さい順に並べ替える機能がありますが、表を書き換えてしまうので利用しないようにしてください。(ソートした結果は、フィルターを解除しても戻らないようですので、設定した場合 **Ctrl**+**Z** で元に戻してください。

(1) 項目名の列を範囲してします。GDP の例では、例えば A1:Y1 (Country から 中央値の列まで) を範囲指定します。

(2) メニューの **データ** → **並べ替えとフィルター** の中の **フィルター**

(3) A1:X1 の右に、 のボタンが表示されます。

■フィルターを使って、行の指定 次に、フィルターで扱う行を制限します (合計や平均の行を扱わないようにします)

(1) 項目名の右の  のボタンをクリックします (ここでは、A1 の右の  ボタンをクリック)

(2) メニューが表示されるので、扱わない行（ここでは、「合計」、「平均」、「空白のセル」のチェックを外します。

(3) をクリック

■抽出 オートフィルターを使うことにより、必要な行だけ抽出することができます。たとえば、2017年のGDPが10000百万ドルより多い行だけ選択してみます。

(1) 基準の列（2017の列）の  のボタンをクリックします

(2) 数値フィルターをクリックします、

(3) 「指定の値より大きい」を選び、10000を入力し、をクリック

■フィルターの解除 使い終わったら、フィルターの解除を行ってください。

(1) メニューの  →  の中の

※ 抽出等をクリアします

(2) メニューの  →  の中の

※フィルターのボタンなどを消します。

※フィルターの機能に、並べかえの機能がありますが、これを利用すると表を書き換えてしまうので、この演習では利用しないようにしましょう。

## 10 IF 関数

### 10.1 IF 関数とは

S6 の売上高評価の表で、「伸び率評価1」の計算式を設定しましょう。これは、伸び率が正だったら 1、0 または負だったら、0 を表示させます。

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図 14 売上高の評価(再掲)

あるときは、こうして、そうでないときは、こうするといったことをしたいときは、IF 関数を使います。使い方は、

IF(条件, 条件を満たしているときすること, 条件を満たしていないときすること)

となります。したがって、

B6: =IF(B4>0,1,0)

とします。その意味は、図 15 のようになります。




図 15 IF 関数の図解

表 2 Excel での大小関係の表記

大小関係	Excel の表記
=	=
≠	<>
>	>
≥	>=
<	<
≤	<=

大小関係などの条件の表記は、数学の表記とは若干異なり、表 2 のようになります。

練習として、伸び率評価 2 の計算式を設定しましょう。伸び率評価 2 は、その支店の伸び率が全体の伸び率（セル G4 の値）より大きいまたは等しければ 1 を、そうでなければ 0 を表示させます。絶対参照またはスピルを使いますので注意して

ください。

## 10.2 講演会参加費

図 16 は、ある講演会の会費の計算をしたものです。この表は、[教科書のページ](#)からダウンロードできます (S9.xlsx)。

会員／非会員 (B 列) 会員と非会員では、参加費が異なります。会員には、「M」と入力されています。

学生／非学生 (C 列) 学生には、参加費が割引 (学生割引) されます。学生には、「S」と入力されています。

パーティ (D 列) 講演会終了後に開かれるパーティに参加する人は、「P」と入力されています。

参加費 (E 列) 参加費を記入します。参加費は変更されることもあるので、C24 もしくは C25 を参照するようにします。

学生割引 (F 列) 学生割引を記入します。学生割引は変更されることもあるので、C26 を参照するようにします。

パーティ代 (G 列) パーティ代を記入します。パーティ代は変更されることもあるので、C27 を参照するようにします。

合計 参加費 - 学生割引 + パーティ代 で計算します。

### ■参加費の計算 (スピル)

E2: =IF(B2:B21="M",C24,C25)

動画: [スピル・参加費の計算 \(音声付き\)](#) ・ [動画: スピル・参加費の計算 \(音声なし\)](#)

■参加費の計算 (絶対参照) E2 のセルの計算式を設定します。参加費は、B2 が、「M」か空白によって異なるので IF 関数を使います。

△ E2: =IF(B2="M",6000,9000)

	A	B	C	D	E	F	G	H
1	NO	会員／非会員	学生／非学生	パーティ	参加費	学生割引	パーティ代	合計
2	1	M	S	P	6,000	3,000	4,000	7,000
3	2		S		9,000	3,000	0	6,000
4	3			P	9,000	0	4,000	13,000
5	4	M		P	6,000	0	4,000	10,000
6	5	M	S		6,000	3,000	0	3,000
7	6	M	S		6,000	3,000	0	3,000
8	7	M	S		6,000	3,000	0	3,000
9	8	M		P	6,000	0	4,000	10,000
10	9				9,000	0	0	9,000
11	10	M			6,000	0	0	6,000
12	11	M		P	6,000	0	4,000	10,000
13	12	M			6,000	0	0	6,000
14	13	M			6,000	0	0	6,000
15	14	M	S	P	6,000	3,000	4,000	7,000
16	15	M	S		6,000	3,000	0	3,000
17	16	M		P	6,000	0	4,000	10,000
18	17			P	9,000	0	4,000	13,000
19	18				9,000	0	0	9,000
20	19	M		P	6,000	0	4,000	10,000
21	20	M		P	6,000	0	4,000	10,000
22				合計	135,000	21,000	40,000	154,000
23								
24		参加費(会員)	6,000					
25		参加費(非会員)	9,000					
26		学生割引	3,000					
27		パーティ	4,000					

図 16 講演会会費

B2が「M」かどうか調べるのに、「B2="M"」とします。文字は、""で囲みます。Mのとき、6000なので、真の時の値を6000とし、偽のときの値を9000としました。しかし、その6000と9000という値は、変更されることがあり、C24とC25を参照するようにします。C24,C25は、下に複写されたときも固定される番地なので、絶対参照にします。したがって、

E2: =IF(B2="M", \$C\$24, \$C\$25)

とし、下方向に複製します。

### 10.3 練習問題 S9-1

S9(参加費の計算)を完成させましょう。

### 10.4 練習問題 S8-2

S8の合否を設定します。合格表示という列に、基準点以上であれば1を、そうでなければ0を表示するようにします。もちろん、合否の列は1箇所だけ計算式を作成して、あとは複製で計算式を設定してください。合格表示の合計を算出すれば合格者数を求めることができます。

基準点を変えて、何人合格になるのか試してみようと思います。つまり、基準点を変えてみて、合格者が何人になるのかシミュレーションを行います。

## 11 表引き (LOOKUP 関数)

コンピュータを利用しているとよくコードを使います。例えば、大学生なら学生コード、商品には商品コードが付けられています。また、ある範囲での対応表から表を引くという作業があります。例えば、所得額から、税額を求めるといったことです。これらの機能を実現するのがVLOOKUP(または、HLOOKUP)という関数です。

JR 南武線(路線図)の運賃計算の例で説明しましょう。図17は、南武線の2つの駅名を入力して、川崎からの距離を求め、その差から、2駅間の距離を求め、それから運賃を求めるといったものです。この表は、[テキストのページ](#)からダウンロードできます(S101.xlsx(運賃計算)、南武線は、電車特定区間の運賃)。

	A	B	C
1		駅名	川崎からの距離
2	乗車駅	稲田堤	
3	降車駅	登戸	
4	2駅間距離 切り上げ前		
5	2駅間距離 切り上げ後		
6	運賃(ICカードを利用)		
7	運賃(切符を利用)		
8			
9	駅名	川崎からの距離	
10	川崎	0	
11	武蔵小杉	7.5	
12	武蔵溝ノ口	12.7	
13	登戸	17.3	
14	稲田堤	20.8	
15	稲城長沼	24.1	
16	府中本町	27.9	
17	分倍河原	28.8	
18	立川	35.5	
19			
20	距離	運賃(IC)	運賃(切符)
21	1	133	140
22	4	154	160
23	7	165	170
24	11	216	220
25	16	302	310
26	21	388	390
27	26	464	470
28	31	550	550
29	36	637	640
30	41		
31			

運賃を  
計算

駅名と川  
崎からの  
距離の対  
応表

距離と  
運賃の  
対応表

図 17 JR 南武線の運賃計算

図 17 の表は 3 つの表からできています。一番上が運賃を計算する表、二番目が駅名と川崎からの距離の対応表、三番目は、距離と運賃の対応表です。手順は次のようになります。

- (1) 乗車駅名から川崎からの距離を求めます (C2).
- (2) 降車駅名から川崎からの距離を求めます (C3).

- (3) 2つの「川崎からの距離」の差をもとめ、小数点未満を切り上げます (C4,C5).  
 (4) 距離と運賃の対応表から運賃をもとめます (C6,C7).

## 11.1 乗車距離から運賃を求める

まず乗車距離から運賃 (C6) を求めてみましょう。仮に乗車距離として C5 に 4 を入力します。C6 を求めるには、距離と運賃の対応表を C5 の 4 について引くことです。乗車距離は 4 なので表の口の行「4 154」の行があてはまり運賃は 154 円になります。

JRの運賃表		Excelの表			
	距離	運賃(IC)	距離	運賃(IC)	
イ	1~3	133	1	133	←1以上4未満のとき133円
ロ	4~6	154	4	154	(距離は整数なので1~3のとき130円)
ハ	7~10	165	7	165	
ニ	11~15	216	11	216	←11以上16未満のとき216円
ホ	16~20	302	16	302	(距離は整数なので11~15のとき216円)
ヘ	21~25	388	21	388	
ト	26~30	464	26	464	
チ	31~35	550	31	550	
リ	36~40	637	36	637	←36以上41未満のとき637円
			41		(距離は整数なので36~40のとき637円)

図 18 JR 運賃表と Excel の VLOOKUP 関数用の表との対応関係

### 11.1.1 対応表の見方

図 18 は、JR 運賃表を Excel の VLOOKUP 用の表には、どのように記述したらよいのかを示しています。Excel の表の各行がどの数値をあてはまるかは、その行の距離の値から、次の行の距離の値までです。正確に言うと対応表の各行にあてはまる数値は、その行の距離から次の行の値未満までです。

距離が 1 の行 (イ) 距離が 1 以上 4 未満の運賃が 133 円であることを示しています。

距離が 4 の行 (ロ) 距離が 4 以上 7 未満の運賃が 154 円であることを示しています

例えば、乗車距離が 22 であったら、「21 388」の行 (へ) にあてはまります。

## 11.2 VLOOKUP 関数

C6 の計算式は、図 19 は、VLOOKUP 関数の図解です。

**C6:** =VLOOKUP(C5,A21:C30,2,TRUE)

VLOOKUP 関数は、VLOOKUP(照合値, 照合範囲, 列位置, 照合方法) の 4 つの引数を書きます。(C5 に何も書いていないとエラーになるので、C5 に仮の 2 駅間距離 (例えば 6) を入力し、この値をいろいろ変えて見てください。)

**照合値** 表引きをする表で照合する値。この例では、乗車距離 (C5)

(距離と運賃の対応表の一番左の列から探す値)

**照合範囲 (対応表の範囲)** 照合する表、ただし、照合値の列は、照合する範囲の一番左の列になるようにします。

**列位置** 照合した行の左から何番目の列を表示するか。一番左の列を 1 列目と数えます。

**照合方法** 完全に一致するもののみを探すのか (FALSE)、下の行と未満のものも探すか (TRUE) を指定します。




図 19 VLOOKUP 関数の書き方

照合方法が TRUE のとき、表引きをする表は、一番左の列をソートキーに昇順(小さい順)にソートされてなくてはなりません。理由は、照合するとき、照合値が、一番左の列の値と次の行の一番左の列の値の間にあるかどうかで、照合するかしないか決めているからです。

**C6:** =VLOOKUP(B5,A21:C30,2,TRUE)

練習として、C7の切符を利用した場合の運賃を計算しましょう。この場合3列目になることに注意しましょう。C5の仮の値をいろいろ変更してみて、それぞれの運賃が正しく表示されているか確認しましょう。

動画: [スピル・Vlookup関数（音声付き）](#) ・ [動画: スピル・Vlookup関数（音声なし）](#)

### 11.3 駅名から距離を求める

次に、駅名から「川崎からの距離」を求める関数を求めましょう。駅名から距離を求めるときは、完全に一致する行のみを探します。したがって、ソートしておく必要はありません。C2の計算式は、B2を照合値として、A10:B18の照合範囲を探し、1列目と等しければ、2列目の値を返すというものです。完全に一致する行のみ探すので、「FALSE」にします。C3も同様の計算式のため、絶対参照もしくはスピルを利用します。

C2とC3は、基本的に同じ計算式で、照合値がB2かB3かの違いで、照合範囲は同じで、列位置、照合方法は一定の値です。

#### ■スピルを利用するとき

**C2:** =VLOOKUP(B2:B3,A10:B18,2,FALSE)

範囲B2:B3（縦1列、2個）の値を照合値をスピルの範囲として使い、照合範囲A10:B18は共通で使います。出力は、照合範囲が縦1列、2個（2行）ですので、出力のスピル範囲もC2:C3の縦1列、2個（2行）になります。

動画: [スピル・駅名から距離（音声付き）](#) ・ [動画: スピル・駅名から距離（音声なし）](#)

■絶対参照を利用するとき 絶対参照を利用するときは、照合範囲が常にA10:B18になるので、**\$A\$10:\$B\$18**のように、絶対参照にし、C2をC3に複製します。

**C2:**      =VLOOKUP(B2,\$A\$10:\$B\$18,2,FALSE)  
 複写元:                      C2  
 複写先:                      C3

## 11.4 2 駅間距離を求める (絶対値, 切り上げ)

2つの駅間の距離を求めるには、2つの駅の川崎からの距離の差を求めます。負の値になることがあるので、絶対値をとります。絶対値をとる関数は、ABS です。まず、2 駅間距離切り上げ前は、

**C4:**      =ABS(C2 - C3)

となります。JR の運賃計算は、小数点以下の端数は切り上げです。そこで、C4 の値を切り上げて、C5 に表示します。

**C5:**      =ROUNDUP(C4,0)

となります。ROUNDUP(A,B) で、A は切り上げる数、B は切り上げる単位を示しています。0 で小数点の位置、1 で 0.1 単位、2 で 0.01 単位、-1 で 10 の単位で切り上げです。同様に、ROUND が四捨五入、ROUNDDOWN が切り捨てです (表 3)。

表 3 切り上げ, 四捨五入, 切り捨て

ROUNDUP(3.14,0) → 4	ROUNDUP(586,-1) → 590	ROUNDUP(16.58,1) → 16.6
ROUND(3.14,0) → 3	ROUND(586,-1) → 590	ROUND(16.58,1) → 16.6
ROUNDDOWN(3.14,0) → 3	ROUNDDOWN(586,-1) → 580	ROUNDDOWN(16.58,1) → 16.5

## 12 練習問題

### 12.1 練習問題 S102-1

図 20 は、小田急ロマンスカーの新宿からの運賃、特急料金の表です(路線図)。B14 に駅名を入力したら、B15 の運賃、B16 の特急料金を計算する式を設定しましょう。この表は、[テキストのページ](#)からダウンロードできます(S102.xlsx(ロマンスカー))。

	A	B	C	
1	新宿からのロマンスカー運賃・特急料金			
2		運賃(切符)	特急料金	
3	向ヶ丘遊園	250	300	
4	新百合ヶ丘	310	410	
5	町田	370	410	
6	相模大野	370	410	
7	本厚木	500	570	
8	秦野	670	620	
9	新松田	780	690	
10	小田原	880	890	
11	箱根湯本	1190	890	
12				
13	新宿からロマンスカーを利用(切符)			
14	降車駅	秦野		
15	運賃			
16	特急料金			
17	合計			

図 20 ロマンスカーの運賃・料金

ヒント：B16 を計算するとき、A3:C11 までを照合範囲として、Vlookup 関数を使います。列位置を設定することで特急料金を表示させます。

## 12.2 練習問題 S4-4

S4.xlsx のモーグルで、「エア点」の計算と「スピード」の計算では、小数第 3 桁以下を切り捨て、小数点以下第 2 位まで求めるように変更しましょう。

## 12.3 練習問題 S103-1

図 22 は、表 21 にしたがって、得点から、S,A+,A,B+,B,C+,C,F の評価を計算する表です。VLOOKUP で使う得点と評価の対応表（照合範囲）を作成し、C2~C11 の計算式を設定しなさい。この表は、[テキストのページ](#)からダウンロードできません (S103.xlsx(成績評価))。

得点	評価
90~100	S
85~89	A+
80~84	A
75~79	B+
70~74	B
65~69	C+
60~64	C
0~59	F

図 21 大学の成績評価

	A	B	C	D
1	名前	得点	評価	
2	田中	100	S	
3	大西	70	B	
4	大久保	30	D	
5	前田	80	A	
6	友田	65	C+	
7	保田	77	B+	
8	大川	78	B+	
9	木村	32	D	
10	村田	54	D	
11	川上	66	C+	
12				

図 22 得点から成績を求める

- VLOOKUP 関数で照合方法が TRUE(行と行との間も検索する場合) のとき、照合値の列（一番左の列）が昇順（小さい順）でソート（並べ替え）されていなくてはなりません。したがって、表 21 に対応する VLOOKUP の照合範囲の表は、表 21 の行の順番とは逆になります。「0 から 59」の行が 1 行目、「90～」の行が最下行になります。

### 13 絶対参照と相対参照の混合とそのスピルでの計算

図 23 は、ネットショッピングの支払いを各通貨別に集計したものです。換算レートは、2012 年 3 月現在のものです。赤の部分は計算式で求める部分です。この表は、[テキストのページ](#)からダウンロードできます (S104.xlsx(ネットショッピング))。

表には、日付と品名、支払った通貨の種類（GBP はイギリスポンド、EUR はユーロ、CHF はスイスフラン、USD はアメリカドルをあらわします。）、その通貨での金額を記入します。表の右側で、通貨毎に集計し、換算レートをかけて、日本円（邦貨）の金額を求めます。日付の入力では「8 月 10 日」は「8/10」と入力します。自動的に、「8 月 10 日」に変換されます。

日付としてうまく表示されない場合や、日付から数値の表示形式に変更するときは、右クリックして、セルの書式設定の表示形式のタブで変更します。

最終的には、図 23 のような表にして、各通貨毎の合計等を求めます。

まず、計算式を設定しましょう。まず、E4 について考えます。

E4 は、もし、この行の通貨が、GBP であつたら、この行の金額の欄の値を表示する。

↓

E4 は、もし、C4 と E3 が等しかったら、D4 を表示し、そうでなかったら、空白を表示する（空白は、ダブルコーテーション「”」を 2 つ重ねます）。

↓

	A	B	C	D	E	F	G	H	I
1	ネットショッピング								
2									
3	日付	品名	通貨	金額	GBP	EUR	CHF	USD	
4	8月10日	洋服	GBP	85	85				
5	8月10日	バッグ	GBP	70	70				
6	8月11日	バッグ	EUR	1000		1000			
7	8月13日	洋服	EUR	500		500			
8	8月13日	バッグ	EUR	600		600			
9	8月14日	書籍	EUR	150		150			
10	8月14日	洋服	CHF	150			150		
11	8月14日	書籍	USD	100				100	
12	8月15日	ソフトウェア	USD	130				130	
13		合計			155	2250	150	230	
14		換算レート			132.15	109.88	91.08	83.41	合計
15		邦貨金額(日本円の金額)			20,483	247,230	13,662	19,184	300,560

図 23 ネットショッピング支払い

E4: =IF(C4=E3,D4,"")

となります。次に、この計算式を複製してみましょう。

複製元:

E4

複製先:

E4:H12

としてみましょう。間違っていることがわかります。

このような場合、2つの方法で解決できます。1つはスピルを使った方法で、もう1つは絶対参照と相対参照の混合させる方法です。

## 13.1 スピルを使った計算

スピルを使っての計算式の場合、E4 の `=IF(C4=E3,D4,"")` の IF 関数の各部分について範囲 E4:H12 の全セルに対して、図 24 のように考えます。範囲 D(出力の範囲)は、範囲 A と範囲 B の値が等しいとき、範囲 C の値を出力します。そうではないとき、""を出力します。

	A	B	C	D	E	F	G	H	I	
1	ネットショッピング									
2					B					
3	日付	品名	通貨	金額	GBP	EUR	CHF	USD	Dの計算式	
4	8月10日	洋服	GBP	85	85				A = B だったら C そうでなかったら "" (空白)	
5	8月10日	バッグ	GBP	70	70					
6	8月11日	バッグ	EUR	1000		1000				
7	8月13日	洋服	EUR	500		500				
8	8月13日	バッグ	EUR	600		600				
9	8月14日	書籍	EUR	150		150				
10	8月14日	洋服	CHF	150			150			
11	8月14日	書籍	USD	100				100		
12	8月15日	ソフトウェア	USD	130				130		
13		合計								E4: =IF(C4:C12=E3:H3,D4:D12,"")
14		換算レート			132.15	109.88	91.08	83.41		合計
15		邦貨金額(日本円の金額)								

図 24 スピルによる計算，ネットショッピング支払い

IF の条件部 この部分は、C 列 (C4:C12 範囲 A) と 3 行目 (E3:H3 範囲 B) の値が等しかったらという条件になるので、`C4:C12=E3:H3` とします。

条件を満たしているとき この部分は、D 列 (D4:D12 範囲 C) の値を表示するので、`D4:D12` とします。

条件を満たしていないとき ("" ) そのまま `""` とします

E4: =IF(C4:C12=E3:H3,D4:D12,"")

とします。

動画: スピル・ネットショッピング (音声付き) ・ 動画: スピル・ネットショッピング (音声なし)

## 13.2 絶対参照の利用 (絶対参照と相対参照の混合を使った計算)

E5,E6 の計算式 (表 4 上部) を見てみましょう。E5 の計算式は、IF(C5=E4,D5,"") になっています。しかし、正しい計算式は、IF(C5=E3,D5,"") です。E6,E7 も表のようになります。原因は、E4 の式中の E3 は、下に複写するとき、数字の部分がが増えてはいけないのに、番地調整機能によって、増えてしまっているからです。

表 4 絶対参照と相対参照の混合

セル	単純に複写したときの計算式	正しい計算式
E4:	=IF(C4=E3,D4,"")	=IF(C4=E3,D4,"")
E5:	=IF(C5=E4,D5,"")	=IF(C5=E3,D5,"")
E6:	=IF(C6=E5,D6,"")	=IF(C6=E3,D6,"")
E7:	=IF(C7=E6,D7,"")	=IF(C7=E3,D7,"")
E4:	=IF(C4=E3,D4,"")	=IF(C4=E3,D4,"")
F4:	=IF(D4=F3,E4,"")	=IF(C4=F3,D4,"")
G4:	=IF(E4=G3,F4,"")	=IF(C4=G3,D4,"")
H4:	=IF(F4=H3,G4,"")	=IF(C4=H3,D4,"")

E4 の計算式を右に複写したものを表 4 下部に示します。赤の部分が間違っています。これは、C4 の部分と D4 の部分の

アルファベットは、横に複写されるときに、増えてはいけないのに番地調整機能により、増えてしまったからです。

複写のとき、番地調整機能を働かせないように「\$」を付けます。上の例では、

E4: =IF(\$C4=E\$3,\$D4,"")

となります。複写しても、Cと3とDは、変化せずいつもCと3とDです。E4を設定したら、他の部分に複写します。

複写元： E4      複写先： E4:H12

### 13.3 日本円への換算等

合計などを計算しましょう。換算レートは、1 海外通貨あたり日本円でいくらかを示しています。例えば、155 ポンド (GBP) 使い、換算レートが 1 ポンド=132.154 円のとときの邦貨金額 (日本円金額) は、155 ポンド × 132.15 円/ポンド = 20,483 円となります。I15 は、邦貨金額の合計です。15 行目はセルの書式設定で、小数点以下を四捨五入しましょう。

通貨記号の欄を変更したり、現在の為替レート ( [朝日新聞デジタル：為替 - 経済・マネー](#) ) に直したりしても正しく動作するか確認しましょう。

## 14 練習問題

### 14.1 1 次関数 3 本のグラフ化

1 次関数  $f(x) = ax + b$  で、 $a$  と  $b$  を変えた 3 本をグラフ化する表を作成しましょう。図 25 左の B2:D3 に 3 本の 1 次関数の  $a$  と  $b$  の値を記入しておきます。ただし、計算式を設定する部分は、次のようにします。

(1) A6 に -10 を入力し、A7 に「=A6+0.5」とします。

- (2) A7 の計算式を A46 まで複写します。
- (3) B6:D46 の計算式：絶対参照を利用しもスピルを利用してよい。
- B6 に絶対参照と相対参照が混在した計算式を設定し、
  - その式を B6:D46 に複写します。
- (4) スピル
- B6:D46 を範囲指定して、まとめて計算式を設定します。
  - x の値の範囲：A6:A46
  - a の値の範囲：B2:D2
  - b の値の範囲：考えてみてください

	A	B	C	D
1		イ	ロ	ハ
2	a(傾き)	1	2	-1
3	b(切片)	1	1	1
4				
5	x	イ	ロ	ハ
6	-10	-9	-19	11




図 25  $y = ax + b$  のグラフ化のための表とグラフ

- (5) 図 25 右のような 3 本のグラフ (散布図 (直線)) を作成しましょう。

(6) それぞれの直線のパラメータである  $a, b$  を変化させたとき、どのように変化するのかを確かめましょう。

	A	B	C	D	E	F	G	H	I
1									
2	日	通話区分	距離	通話時間 (秒)	10円で話せる時間	度数 (切り上げ前)	度数 (切り上げ後)	金額	
3	6月10日	1	35	588	45	13.06666667	14	140	
4	6月11日	2	48	627	60	10.45	11	110	
5	6月12日	3	165	37	45	0.822222222	1	10	
6	6月12日	3	165	156	45	3.466666667	4	40	
7	6月13日	2	23	251	60	4.183333333	5	50	
8	6月15日	1	18	1028	90	11.42222222	12	120	
9							合計	470	
10									
11									
12	料金表 (10円で話せる時間)								
13	距離	1 平日昼間	2 夜・休日昼夜	3 深夜早朝					
14	0	90	90	120					
15	20	60	60	75					
16	30	45	60	75					
17	60	26	45	60					
18	100	20	26	45					
19	170	20	26	36					

図 26 電話料金の計算

## 14.2 練習問題 S105-1

図 26 は、ある架空の電話会社の電話料金の計算表です。この表は、[テキストのページ](#)からダウンロードできます (S105.xlsx(電話料金))。

回答は、絶対参照と相対参照を混在させた式で設定してください。スプिलでは、スプिल範囲の値に計算式を追加できないよ

うです。

- 10円で話せる時間は、通話区分、距離で決まります。
- 通話区分は3区分：「平日の昼間」(区分1)、「平日の夜または休日の昼夜」(区分2)、「平日、休日の深夜」(区分3)
- 距離は、通話先までの距離で、表の14行目の0は、0km(隣接)以上20km未満を表しています。また、19行目の距離170の行は、170km以上すべてを表しています。
- たとえば、通話区分が2で、70kmの場合、60の行の通話区分2の列(3列目)の場合の10円で話せる秒数は45秒になります。
- 料金は、まず度数を、

$$\text{度数} = \frac{\text{通話時間(秒)}}{10\text{円で話せる秒数}}$$

で計算し(F列)、小数点以下を切り上げます(G列、切り上げの計算方法は、表3を参照)。

- 1度数は10円なので、度数の列(G列)に10をかけたものが電話料金になります。

では、図26の上方のエリアを使って電話料金の計算表を作成しましょう。

ヒント

- 「10円で話せる秒数」は、距離と通話区分を使って、料金表(A14:D19)からVLOOKUPを使って探します。そのとき、通話区分が1のとき2列目、2のとき3列目...であることから、VLOOKUPの列位置は、通話区分のセルの値を使った計算式になります(通話区分の値に1加えた数が列番号になっています)。
- 「度数(切り上げ前)」は、単純に、 $\frac{\text{通話時間(秒)}}{10\text{円で話せる秒数}}$ を計算したものです。
- 「度数(切り上げ後)」は、「度数(切り上げ前)」の値を切り上げたものです。

完成したら、通話区域、通話の種類や時間を変えてみて、正しく計算されているかどうかチェックしましょう。